Chirp-free solitons in dissipative systems: variational approximation and issue of the soliton energy scalability

V.L.Kalashnikov
Instut fuer Photonik, TU Wien
e-mail: kalashnikov@tuwien.ac.at
30.05.2014

Sources

pdf-version

Mathematica notebook

Introduction

Variational approximation is based on the truncation of infinitely dimensional phase space of a dynamical system by the means of trial solution anzatz with the subsequent variational procedure for the finite set of anzatz parameters (see Refs.)

References:

D. Anderson, et al., Pramana J. Phys. 57, 917–936 (2001); B.A.Malomed, in: Progress in Optics **43**, 71 (2002); A. Ankiewicz, et al., Opt. Fiber Technology **13**, 91 (2007); S.K.Turitsyn, et al., Phys, Reports **521**, Issue 4 (2012).

We will consider a realization of this procedure on the examples of chirp-free solitons developing in the one-dimensional distributed dissipative systems, which simulate a mode-locked laser operating in the anomalous dispersion regime and correspond to the complex nonlinear Ginzburg-Landau equation [N.N.Akhmediev, A.Ankiewicz (Eds.), Dissipative Solitons (Springer, 2005)]. The instantaneous response of dissipative system can be considered as an approximated model for the so-called Kerr-lens mode locking mechanism [R. Paschotta. *Encyclopedia of Laser Physics and Technology* (John Wiley & Sons, 2008)]. We will attend especially to the issue of the soliton energy scalability, which is of interest for high-field generation directly from a laser [T. Suedmeyer, et al., Nature Photon. **2**, 599 (2002)].

```
Needs["VariationalMethods`"]
```

True soliton

The modern methods of ultra short pulse generation use the so-called dispersion compensation technique [F. X. Kaertner (Ed.). Few-Cycle Laser Pulse Generation and Its Applications (Springer-Verlag, 2004)]. The master

equation modelling this technique is the nonlinear Schroedinger equation [Y.S. Kivshar, G.P. Agrawal. *Optical Solitons: From Fibers To Photonic Crystals* (Elsevier, 2003)]. Lagrangian density for the nonlinear Schroedinger equation is (x is the propagation coordinate, t is the local time, β is the net group-delay dispersion, and γ is the self-phase modulation coefficient)

$$L = \frac{i}{2}\left(A1[x,t] * \partial_x A[x,t] - A[x,t] * \partial_x A1[x,t]\right) - \frac{1}{2}\gamma\, A1[x,t]^2 * A[x,t]^2 + \frac{1}{2}\beta\, \partial_t A[x,t] * \partial_t A1[x,t]$$

$$-\frac{1}{2}\gamma\, A[x,t]^2\, A1[x,t]^2 + \frac{1}{2}\beta\, A^{(0,1)}[x,t]\, A1^{(0,1)}[x,t] + \frac{1}{2} i \left(A1[x,t]\, A^{(1,0)}[x,t] - A[x,t]\, A1^{(1,0)}[x,t]\right)$$

Corresponding Euler-Lagrange equations result in the nonlinear Schroedinger equation for the field *A* and the complex-conjugated field *A1*

```
Expand[EulerEquations[L, {A[x, t], A1[x, t]}, {x, t}]]
```

$$\left\{-\gamma\, A[x,t]\, A1[x,t]^2 - \frac{1}{2}\beta\, A1^{(0,2)}[x,t] - i\, A1^{(1,0)}[x,t] == 0,\right.$$
$$\left. -\gamma\, A[x,t]^2\, A1[x,t] - \frac{1}{2}\beta\, A^{(0,2)}[x,t] + i\, A^{(1,0)}[x,t] == 0\right\}$$

Substitution of the trial *sech*-function with varying amplitude α, width T, and phase ϕ gives the Lagrangian density

$$A[x\_, t\_] := \alpha[x]\, \mathrm{Sech}\left[\frac{t}{T[x]}\right] \mathrm{Exp}[i\, \phi[x]]$$

$$A1[x\_, t\_] := \alpha[x]\, \mathrm{Sech}\left[\frac{t}{T[x]}\right] \mathrm{Exp}[-i\, \phi[x]]$$

```
L2 = L // FullSimplify
```

$$\frac{1}{2\, T[x]^2}\, \mathrm{Sech}\left[\frac{t}{T[x]}\right]^2 \alpha[x]^2 \left(\beta - \mathrm{Sech}\left[\frac{t}{T[x]}\right]^2 \left(\beta + \gamma\, T[x]^2\, \alpha[x]^2\right) - 2\, T[x]^2\, \phi'[x]\right)$$

The corresponding reduced Lagrangian is

```
L2reduced = Integrate[L2, {t, -∞, ∞}, Assumptions →
   t ∈ Reals && T[x] ∈ Reals && T[x] > 0 && α[x] ∈ Reals && α[x] > 0 && ϕ[x] ∈ Reals]
```

$$\frac{\alpha[x]^2 \left(\beta - 2\, T[x]^2 \left(\gamma\, \alpha[x]^2 + 3\, \phi'[x]\right)\right)}{3\, T[x]}$$

Hence, the Euler-Lagrange equations are

```
Euler1 = EulerEquations[L2reduced, {α[x], T[x], ϕ[x]}, x]
```

$$\left\{\frac{2\,\alpha[x]\,\left(\beta-2\,T[x]^2\,\left(2\,\gamma\,\alpha[x]^2+3\,\phi'[x]\right)\right)}{3\,T[x]}==0,\right.$$
$$\left.-\frac{\alpha[x]^2\,\left(\beta+2\,T[x]^2\,\left(\gamma\,\alpha[x]^2+3\,\phi'[x]\right)\right)}{3\,T[x]^2}==0,\;2\,\alpha[x]\,\left(\alpha[x]\,T'[x]+2\,T[x]\,\alpha'[x]\right)==0\right\}$$

```
eq1 = Euler1[[1]]
eq2 = Euler1[[2]]
eq3 = Euler1[[3]]
```

$$\frac{2\,\alpha[x]\,\left(\beta-2\,T[x]^2\,\left(2\,\gamma\,\alpha[x]^2+3\,\phi'[x]\right)\right)}{3\,T[x]}==0$$

$$-\frac{\alpha[x]^2\,\left(\beta+2\,T[x]^2\,\left(\gamma\,\alpha[x]^2+3\,\phi'[x]\right)\right)}{3\,T[x]^2}==0$$

$$2\,\alpha[x]\,\left(\alpha[x]\,T'[x]+2\,T[x]\,\alpha'[x]\right)==0$$

After some simplifications, one has

```
eq4 = Simplify[eq1, {T[x] ≠ 0, α[x] ≠ 0}]
eq5 = Simplify[eq2, {T[x] ≠ 0, α[x] ≠ 0}]
eq6 = Expand[eq3]
```

$$T[x]^2\,\left(4\,\gamma\,\alpha[x]^2+6\,\phi'[x]\right)==\beta$$

$$\beta+2\,T[x]^2\,\left(\gamma\,\alpha[x]^2+3\,\phi'[x]\right)==0$$

$$2\,\alpha[x]^2\,T'[x]+4\,T[x]\,\alpha[x]\,\alpha'[x]==0$$

Final equations are (note that *eq6* is the energy conservation law)

```
Solve[{eq4, eq5}, {α[x], ϕ'[x]}] // FullSimplify
Solve[eq6, T'[x]] // FullSimplify
```

$$\left\{\left\{\alpha[x]\to-\frac{\sqrt{\beta}}{\sqrt{\gamma}\,T[x]},\;\phi'[x]\to-\frac{\beta}{2\,T[x]^2}\right\},\;\left\{\alpha[x]\to\frac{\sqrt{\beta}}{\sqrt{\gamma}\,T[x]},\;\phi'[x]\to-\frac{\beta}{2\,T[x]^2}\right\}\right\}$$

$$\left\{\left\{T'[x]\to-\frac{2\,T[x]\,\alpha'[x]}{\alpha[x]}\right\}\right\}$$

These equations represent the so-called Schroedinger soliton when both $\partial_x T$ and $\partial_x \alpha$=0.

Kantorovitch's method

1) Perfectly saturable absorber

Firstly, let’s consider the so-called perfectly saturable absorber which loss coefficient decreases monotonically with power. The dissipative term in this case is read as: Q= $-i\Gamma A + i\frac{\rho}{1+\sigma\int_{-\infty}^{\infty} AA1\, dt}(A + \tau\,\partial_{t,t} A) + \frac{i\mu\zeta AA1}{1+\zeta AA1} A$ (Γ is the unsaturable loss, ρ is the gain for a small signal, σ is the inverse energy of gain saturation, τ is the squared inverse gain-bandwidth, μ is the saturable absorber modulation depth, ζ is the inverse loss saturation power). The Q-term and the functional derivatives of the complex-conjugated field are

```
Q = -ⅈ Γ A[x, t] +
  ⅈ ρ / (1 + σ Integrate[A[x, t] * A1[x, t], {t, -∞, ∞}, Assumptions → t ∈ Reals &&
        T[x] ∈ Reals && T[x] > 0 && α[x] ∈ Reals && α[x] > 0 && ϕ[x] ∈ Reals]) *
   (A[x, t] + τ ∂t,t A[x, t]) + ⅈ μ (ζ A[x, t] * A1[x, t]) A[x, t] / (1 + ζ A[x, t] * A1[x, t])
u1 = VariationalD[A1[x, t], α[x], x]
u2 = VariationalD[A1[x, t], T[x], x]
u3 = VariationalD[A1[x, t], ϕ[x], x]
```

$$-i\, e^{i\phi[x]}\,\Gamma\,\mathrm{Sech}\left[\frac{t}{T[x]}\right]\alpha[x] + \frac{i\, e^{i\phi[x]}\,\zeta\,\mu\,\mathrm{Sech}\left[\frac{t}{T[x]}\right]^3\alpha[x]^3}{1+\zeta\,\mathrm{Sech}\left[\frac{t}{T[x]}\right]^2\alpha[x]^2} + \left(i\,\rho\left(e^{i\phi[x]}\,\mathrm{Sech}\left[\frac{t}{T[x]}\right]\alpha[x] + \tau\left(-\frac{e^{i\phi[x]}\,\mathrm{Sech}\left[\frac{t}{T[x]}\right]^3\alpha[x]}{T[x]^2} + \frac{e^{i\phi[x]}\,\mathrm{Sech}\left[\frac{t}{T[x]}\right]\mathrm{Tanh}\left[\frac{t}{T[x]}\right]^2\alpha[x]}{T[x]^2}\right)\right)\right)\Big/\left(1+2\,\sigma\,T[x]\,\alpha[x]^2\right)$$

$$e^{-i\phi[x]}\,\mathrm{Sech}\left[\frac{t}{T[x]}\right]$$

$$\frac{e^{-i\phi[x]}\,t\,\mathrm{Sech}\left[\frac{t}{T[x]}\right]\mathrm{Tanh}\left[\frac{t}{T[x]}\right]\alpha[x]}{T[x]^2}$$

$$-i\, e^{-i\phi[x]}\,\mathrm{Sech}\left[\frac{t}{T[x]}\right]\alpha[x]$$

The "source" term within the framework of the Kantarovitch' s method consists of the *f1* -, *f2* - and *f3* - functions, which are integrated products of the *Q*-term and the variational derivatives of *A1* (i.e. the *u1*-, *u2*- and *u3*-functions):

```
f1 = Integrate[Expand[Q * u1], {t, -∞, ∞}, Assumptions → t ∈ Reals && T[x] ∈ Reals &&
    T[x] > 0 && α[x] ∈ Reals && α[x] > 0 && ϕ[x] ∈ Reals && ζ > 0 && ζ ∈ Reals]
f2 = Integrate[Expand[Q * u2], {t, -∞, ∞}, Assumptions → t ∈ Reals && T[x] ∈ Reals &&
    T[x] > 0 && α[x] ∈ Reals && α[x] > 0 && ϕ[x] ∈ Reals && ζ > 0 && ζ ∈ Reals]
f3 = Integrate[Expand[Q * u3], {t, -∞, ∞}, Assumptions → t ∈ Reals && T[x] ∈ Reals &&
    T[x] > 0 && α[x] ∈ Reals && α[x] > 0 && ϕ[x] ∈ Reals && ζ > 0 && ζ ∈ Reals]
```

$$-i\left(\frac{2\rho\,\tau\,\alpha[x]}{3\,T[x]+6\,\sigma\,T[x]^2\,\alpha[x]^2}+T[x]\left(-\frac{\mu\,\mathrm{Log}\left[\frac{1+\zeta\,\alpha[x]^2-\alpha[x]\sqrt{\zeta\left(1+\zeta\,\alpha[x]^2\right)}}{1+\zeta\,\alpha[x]^2+\alpha[x]\sqrt{\zeta\left(1+\zeta\,\alpha[x]^2\right)}}\right]}{\sqrt{\zeta\left(1+\zeta\,\alpha[x]^2\right)}}+\alpha[x]\left(2\,\Gamma-2\,\mu-\frac{2\,\rho}{1+2\,\sigma\,T[x]\,\alpha[x]^2}\right)\right)\right)$$

$$-\frac{1}{12}\,i\left(\frac{1}{\zeta}\,\mu\left(\pi^2+6\,\mathrm{PolyLog}\left[2,1\Big/\left(-1-2\,\zeta\,\alpha[x]^2+2\,\alpha[x]\sqrt{\zeta\left(1+\zeta\,\alpha[x]^2\right)}\right)\right]+6\,\mathrm{PolyLog}\left[2,-1\Big/\left(1+2\,\zeta\,\alpha[x]^2+2\,\alpha[x]\sqrt{\zeta\left(1+\zeta\,\alpha[x]^2\right)}\right)\right]\right)+4\,\alpha[x]^2\left(3\,\Gamma-3\,\mu-\frac{3\,\rho}{1+2\,\sigma\,T[x]\,\alpha[x]^2}-\frac{\rho\,\tau}{T[x]^2\left(1+2\,\sigma\,T[x]\,\alpha[x]^2\right)}\right)\right)$$

$$\alpha[x]\left(-\frac{2\rho\,\tau\,\alpha[x]}{3\,T[x]+6\,\sigma\,T[x]^2\,\alpha[x]^2}+T[x]\left(\frac{\mu\,\mathrm{Log}\left[\frac{1+\zeta\,\alpha[x]^2-\alpha[x]\sqrt{\zeta\left(1+\zeta\,\alpha[x]^2\right)}}{1+\zeta\,\alpha[x]^2+\alpha[x]\sqrt{\zeta\left(1+\zeta\,\alpha[x]^2\right)}}\right]}{\sqrt{\zeta\left(1+\zeta\,\alpha[x]^2\right)}}+2\,\alpha[x]\left(-\Gamma+\mu+\frac{\rho}{1+2\,\sigma\,T[x]\,\alpha[x]^2}\right)\right)\right)$$

Thus, the "driven" Euler-Lagrange equations are (NB: the right-hand sides, or "source" terms, are **2 Re[*f*]**):

```
fun1 = eq1[[1]] == 0
fun2 = eq2[[1]] == 0
fun3 = eq3[[1]] == 2 f3
```

$$\frac{2\,\alpha[x]\,\left(\beta - 2\,T[x]^2\,\left(2\,\gamma\,\alpha[x]^2 + 3\,\phi'[x]\right)\right)}{3\,T[x]} == 0$$

$$-\frac{\alpha[x]^2\,\left(\beta + 2\,T[x]^2\,\left(\gamma\,\alpha[x]^2 + 3\,\phi'[x]\right)\right)}{3\,T[x]^2} == 0$$

$$2\,\alpha[x]\,\left(\alpha[x]\,T'[x] + 2\,T[x]\,\alpha'[x]\right) == 2\,\alpha[x]\,\left(-\frac{2\,\rho\,\tau\,\alpha[x]}{3\,T[x] + 6\,\sigma\,T[x]^2\,\alpha[x]^2} + T[x]\,\left(\frac{\mu\,\mathrm{Log}\left[\frac{1+\zeta\,\alpha[x]^2-\alpha[x]\,\sqrt{\zeta\,\left(1+\zeta\,\alpha[x]^2\right)}}{1+\zeta\,\alpha[x]^2+\alpha[x]\,\sqrt{\zeta\,\left(1+\zeta\,\alpha[x]^2\right)}}\right]}{\sqrt{\zeta\,\left(1+\zeta\,\alpha[x]^2\right)}} + 2\,\alpha[x]\,\left(-\Gamma + \mu + \frac{\rho}{1 + 2\,\sigma\,T[x]\,\alpha[x]^2}\right)\right)\right)$$

Soliton (steady-state chirp-free pulse)

The soliton parameters are x-independent (except the phase ϕ)

```
fun3 /. {α'[x] → 0, T'[x] → 0}
fun3b = (((%[[2]]) / 2 / α[x]) /. {α[x] → α, T[x] → T}) == 0
Solve[{(fun1 /. {α[x] → α, T[x] → T}), (fun2 /. {α[x] → α, T[x] → T})}, {ϕ'[x], T}]
```

$$0 == 2\,\alpha[x]\left(-\frac{2\,\rho\,\tau\,\alpha[x]}{3\,T[x]+6\,\sigma\,T[x]^2\,\alpha[x]^2}+T[x]\left(\frac{\mu\,\mathrm{Log}\left[\frac{1+\zeta\,\alpha[x]^2-\alpha[x]\,\sqrt{\zeta\,\left(1+\zeta\,\alpha[x]^2\right)}}{1+\zeta\,\alpha[x]^2+\alpha[x]\,\sqrt{\zeta\,\left(1+\zeta\,\alpha[x]^2\right)}}\right]}{\sqrt{\zeta\,\left(1+\zeta\,\alpha[x]^2\right)}}+2\,\alpha[x]\left(-\Gamma+\mu+\frac{\rho}{1+2\,\sigma\,T[x]\,\alpha[x]^2}\right)\right)\right)$$

$$-\frac{2\,\alpha\,\rho\,\tau}{3\,T+6\,T^2\,\alpha^2\,\sigma}+T\left(2\,\alpha\left(-\Gamma+\mu+\frac{\rho}{1+2\,T\,\alpha^2\,\sigma}\right)+\frac{\mu\,\mathrm{Log}\left[\frac{1+\alpha^2\,\zeta-\alpha\,\sqrt{\zeta\,\left(1+\alpha^2\,\zeta\right)}}{1+\alpha^2\,\zeta+\alpha\,\sqrt{\zeta\,\left(1+\alpha^2\,\zeta\right)}}\right]}{\sqrt{\zeta\,\left(1+\alpha^2\,\zeta\right)}}\right) == 0$$

$$\left\{\left\{\phi'[x]\to-\frac{\alpha^2\,\gamma}{2},\ T\to-\frac{\sqrt{\beta}}{\alpha\,\sqrt{\gamma}}\right\},\ \left\{\phi'[x]\to-\frac{\alpha^2\,\gamma}{2},\ T\to\frac{\sqrt{\beta}}{\alpha\,\sqrt{\gamma}}\right\}\right\}$$

Thus, one has solutions for the phase slip $\partial_x\,\phi$ and the soliton width T. The equation for the soliton amplitude remains

```
fun3b
```

$$-\frac{2\,\alpha\,\rho\,\tau}{3\,T+6\,T^2\,\alpha^2\,\sigma}+T\left(2\,\alpha\left(-\Gamma+\mu+\frac{\rho}{1+2\,T\,\alpha^2\,\sigma}\right)+\frac{\mu\,\mathrm{Log}\left[\frac{1+\alpha^2\,\zeta-\alpha\,\sqrt{\zeta\,\left(1+\alpha^2\,\zeta\right)}}{1+\alpha^2\,\zeta+\alpha\,\sqrt{\zeta\,\left(1+\alpha^2\,\zeta\right)}}\right]}{\sqrt{\zeta\,\left(1+\alpha^2\,\zeta\right)}}\right) == 0$$

The crucial step is to renormalize the inverse squared gain-bandwidth to that for a saturated gain. That is $\tau_{\mathrm{new}} = \frac{\rho\,\tau_{\mathrm{old}}}{1+2\,\alpha^2\,T\,\sigma}$. Then, let us introduce the saturated net gain coefficient $\Sigma=\frac{\rho}{1+\alpha^2\,\sigma}-\Gamma$. This term can be expanded in the vicinity of 0, so that the leading term can be expressed in the form of $\delta\frac{E}{\Xi}$, where δ is the parameter expressed through the initial gain and the net-loss [V.L.Kalashnikov, et al., Applied Physics B **83**, 503 (2006)], E is the soliton energy, and Ξ is the cw-energy. Then, the modified equation is

```
fun3c = ((((-2τ/(3T^2) + 2Σ + μ(2 + Log[(1+α^2 ξ-α Sqrt[ξ(1+α^2 ξ)])/(1+α^2 ξ+α Sqrt[ξ(1+α^2 ξ)])]/Sqrt[ξ α^2 (1+α^2 ξ)]))) /. Σ → δ (2α^2 T)/Ξ) /. T → Sqrt[β]/(α Sqrt[γ])) == 0
```

$$\frac{4\,\alpha\sqrt{\beta}\;\delta}{\sqrt{\gamma}\;\Xi}-\frac{2\,\alpha^2\,\gamma\,\tau}{3\,\beta}+\mu\left(2+\frac{\mathrm{Log}\left[\frac{1+\alpha^2\,\xi-\alpha\sqrt{\xi\,\left(1+\alpha^2\,\xi\right)}}{1+\alpha^2\,\xi+\alpha\sqrt{\xi\,\left(1+\alpha^2\,\xi\right)}}\right]}{\sqrt{\alpha^2\,\xi\,\left(1+\alpha^2\,\xi\right)}}\right)==0$$

- Let's the power to be normalized to ζ and $c=\tau\gamma/\beta\zeta$ is some new control parameter. Ξ is the cw-energy normalized to $\zeta/\sqrt{\tau}$. Then

```
fun3d = 4αδ/(Sqrt[c] Ξ) - 2α^2 c/3 + μ(2 + Log[(1+α^2-α Sqrt[1+α^2])/(1+α^2+α Sqrt[1+α^2])]/Sqrt[α^2 (1+α^2)]) == 0

fun3e = 2Σ - 2α^2 c/3 + μ(2 + Log[(1+α^2-α Sqrt[1+α^2])/(1+α^2+α Sqrt[1+α^2])]/Sqrt[α^2 (1+α^2)]) == 0
```

$$-\frac{2\,c\,\alpha^2}{3}+\frac{4\,\alpha\,\delta}{\sqrt{c}\;\Xi}+\mu\left(2+\frac{\mathrm{Log}\left[\frac{1+\alpha^2-\alpha\sqrt{1+\alpha^2}}{1+\alpha^2+\alpha\sqrt{1+\alpha^2}}\right]}{\sqrt{\alpha^2\,\left(1+\alpha^2\right)}}\right)==0$$

$$-\frac{2\,c\,\alpha^2}{3}+2\,\Sigma+\mu\left(2+\frac{\mathrm{Log}\left[\frac{1+\alpha^2-\alpha\sqrt{1+\alpha^2}}{1+\alpha^2+\alpha\sqrt{1+\alpha^2}}\right]}{\sqrt{\alpha^2\,\left(1+\alpha^2\right)}}\right)==0$$

In the last equation, we restored Σ because the marginally stable solution (i.e. the solutions with Σ=0) will be considered hereinafter. As a result, one can obtain the dependencies of the normalized soliton energy and width on the c-parameter. The energy is normalized to $\zeta/\sqrt{\tau}$, the width is normalized to $\sqrt{\tau}$. It should be noted, that τ is the renormalized parameter taking into account the gain saturation, that is the inverse squared gain-bandwidth multiplied by the saturated gain coefficient. But the last equals approximately to Γ. Hence $\tau_{new} = \tau_{old}\times\Gamma$. If the main source of spectral dissipation is a spectral filter, then the gain saturation does not contribute to this process and $\tau_{new} = \tau_{old}$.
Below, we plot the master diagram c vs. E for Σ=0 (the solitons are stable below the curve corresponding to Σ=0), and the dependence of the half-amplitude pulse duration on the energy.

```
startc = 0.000001;(*scaling of the c-parameter starts from this value*)
finc = 0.1;(*scaling of the c-parameter stops at this value*)
steps = (finc - startc) / startc;(*step-size*)
```

```
sol1 = Table[
   {c, FindRoot[fun3e /. {μ → 0.02, Σ → 0}, {α, 1000}, MaxIterations → 500][[1]][[2]]},
   {c, startc, finc, startc}];(*saturable loss coefficient μ=0.02*)
(*c-scaling and search of soliton amplitudes*)
par = Table[sol1[[i]][[1]], {i, 1, steps}];(*table of c-values*)
sol2 = Table[(1/(α Sqrt[c])) /. {α → sol1[[i]][[2]], c → par[[i]]}, {i, 1, steps}];
(*table of soliton width*)
sol3 = Table[2 * (sol1[[i]][[2]])^2 * sol2[[i]], {i, 1, steps}];
(*table of soliton energies*)

sol1b = Table[
   {c, FindRoot[fun3e /. {μ → 0.04, Σ → 0}, {α, 1000}, MaxIterations → 500][[1]][[2]]},
   {c, startc, finc, startc}];(*saturable loss coefficient μ=0.04*)
sol2b = Table[(1/(α Sqrt[c])) /. {α → sol1b[[i]][[2]], c → par[[i]]}, {i, 1, steps}];
(*table of soliton width*)
sol3b = Table[2 * (sol1b[[i]][[2]])^2 * sol2b[[i]], {i, 1, steps}];
(*table of soliton energies*)
```

FindRoot::lstol :

The line search decreased the step size to within tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the merit function. You may need more than MachinePrecision digits of working precision to meet these tolerances. ≫

FindRoot::lstol :

The line search decreased the step size to within tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the merit function. You may need more than MachinePrecision digits of working precision to meet these tolerances. ≫

FindRoot::lstol :

The line search decreased the step size to within tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the merit function. You may need more than MachinePrecision digits of working precision to meet these tolerances. ≫

General::stop : Further output of FindRoot::lstol will be suppressed during this calculation. ≫

FindRoot::lstol :

The line search decreased the step size to within tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the merit function. You may need more than MachinePrecision digits of working precision to meet these tolerances. ≫

FindRoot::lstol :

The line search decreased the step size to within tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the merit function. You may need more than MachinePrecision digits of working precision to meet these tolerances. ≫

FindRoot::lstol :
The line search decreased the step size to within tolerance specified by AccuracyGoal and PrecisionGoal but was unable to find a sufficient decrease in the merit function. You may need more than MachinePrecision digits of working precision to meet these tolerances. ≫

General::stop : Further output of FindRoot::lstol will be suppressed during this calculation. ≫

- Asymptotical behavior of the stability border for $E\to\infty$ can be approximated by the dimensionless law: $c \approx \frac{3.5\sqrt{\mu}}{\Xi}$

```
Show[ListPlot[Table[{Log10[sol3[[i]]], Log10[par[[i]]]}, {i, 1, steps}],
  PlotRange → {{-3, 6}, {-6, -1}}, Joined → True, AxesLabel → {lg[Energy], lg[c]}],
 ListPlot[Table[{Log10[sol3b[[i]]], Log10[par[[i]]]}, {i, 1, steps}],
  PlotRange → {{-3, 6}, {-6, -1}}, Joined → True, AxesLabel → {lg[Energy], lg[c]}],
 Plot[Log10[3.5 Sqrt[μ] / Ξ] /. {μ → 0.02, Ξ → 10^x}, {x, -3, 6}],
 Plot[Log10[3.5 Sqrt[μ] / Ξ] /. {μ → 0.04, Ξ → 10^x}, {x, -3, 6}]]
 (*stability threshold: maximum c vs. energy*)
Show[
 ListPlot[Table[{Log10[sol3[[i]]], sol2[[i]] * 2 * N[ArcSech[1/Sqrt[2]]]}, {i, 1, steps}],
  PlotRange → {{-3, 6}, {1, 100}}, Joined → True, AxesLabel → {lg[Energy], T}],
 ListPlot[Table[{Log10[sol3b[[i]]], sol2b[[i]] * 2 * N[ArcSech[1/Sqrt[2]]]}, {i, 1, steps}],
  PlotRange → {{-3, 6}, {1, 100}}, Joined → True, AxesLabel → {lg[Energy], T}]]
 (*FWHM-width of soliton vs. energy at the threshold*)
```





- Thus, we obtained the scaling law for the asymptotic energy of soliton along the stability border (zero isogain):

$$E \approx \frac{3.5\sqrt{\mu}\ \beta}{\sqrt{\tau}\ \gamma} \tag{1}$$

- and the corresponding asymptotic pulse amplitude and width are

```
Solve[{2 α^2 T == 3.5 Sqrt[μ] β/(Sqrt[τ] γ), T == Sqrt[β]/(α Sqrt[γ])}, {α, T}] // FullSimplify
```

$$\left\{\left\{\alpha \to \frac{1.75\sqrt{\beta}\ \sqrt{\mu}}{\sqrt{\gamma}\ \sqrt{\tau}},\ T \to \frac{0.571429\sqrt{\tau}}{\sqrt{\mu}}\right\}\right\}$$

- NB: the asymptotic soliton width is defined by only inverse gainbandwidth and modulation depth.

2) Cubic-quintic nonlinear gain

Now, let's consider a "driven" system with saturable gain, loss, spectral dissipation and cubic-quintic nonlinear gain term: Q= $-i\Gamma A + i\frac{\rho}{1+\sigma\int_{-\infty}^{\infty} AA1\, dt}(A + \tau\,\partial_{t,t} A) + i\kappa\left(AA1 - \zeta (AA1)^2\right) A$. Here Γ is the net-loss for a small signal, ρ is the initial gain, σ is the inverse energy of gain saturation, τ is the squared inverse gain-bandwidth, μ is the saturable absorber modulation depth, κ is the inverse loss saturation power. The loss saturation is saturable by-turn (ζ-term), so that the minimum of saturable loss (modulation depth coefficient μ)

```
(κ x - κ ζ x^2) /. Solve[∂x (κ x - κ ζ x^2) == 0, x][[1]][[1]]
```

$$\frac{\kappa}{4\,\zeta}$$

is reached, when the power becomes

```
Solve[∂x (κ x - κ ζ x^2) == 0, x][[1]][[1]][[2]]
```

$$\frac{1}{2\,\zeta}$$

The Q-term and the functional derivatives of the complex-conjugated field are

```
Q = -ⅈ Γ A[x, t] +
  ⅈ ρ / (1 + σ Integrate[A[x, t] * A1[x, t], {t, -∞, ∞}, Assumptions → t ∈ Reals &&
        T[x] ∈ Reals && T[x] > 0 && α[x] ∈ Reals && α[x] > 0 && ϕ[x] ∈ Reals]) *
   (A[x, t] + τ ∂t,t A[x, t]) + ⅈ κ (A[x, t] * A1[x, t] - ζ * (A[x, t] * A1[x, t])^2) A[x, t]
u1 = VariationalD[A1[x, t], α[x], x]
u2 = VariationalD[A1[x, t], T[x], x]
u3 = VariationalD[A1[x, t], ϕ[x], x]
```

$$-i\, e^{i\phi[x]}\, \Gamma\, \mathrm{Sech}\left[\frac{t}{T[x]}\right] \alpha[x] + i\, e^{i\phi[x]}\, \kappa\, \mathrm{Sech}\left[\frac{t}{T[x]}\right] \alpha[x] \left(\mathrm{Sech}\left[\frac{t}{T[x]}\right]^2 \alpha[x]^2 - \zeta\, \mathrm{Sech}\left[\frac{t}{T[x]}\right]^4 \alpha[x]^4\right) + \left(i\, \rho \left(e^{i\phi[x]}\, \mathrm{Sech}\left[\frac{t}{T[x]}\right] \alpha[x] + \tau \left(-\frac{e^{i\phi[x]}\, \mathrm{Sech}\left[\frac{t}{T[x]}\right]^3 \alpha[x]}{T[x]^2} + \frac{e^{i\phi[x]}\, \mathrm{Sech}\left[\frac{t}{T[x]}\right] \mathrm{Tanh}\left[\frac{t}{T[x]}\right]^2 \alpha[x]}{T[x]^2}\right)\right)\right) \Big/ \left(1 + 2\, \sigma\, T[x]\, \alpha[x]^2\right)$$

$$e^{-i\phi[x]}\, \mathrm{Sech}\left[\frac{t}{T[x]}\right]$$

$$\frac{e^{-i\phi[x]}\, t\, \mathrm{Sech}\left[\frac{t}{T[x]}\right] \mathrm{Tanh}\left[\frac{t}{T[x]}\right] \alpha[x]}{T[x]^2}$$

$$-i\, e^{-i\phi[x]}\, \mathrm{Sech}\left[\frac{t}{T[x]}\right] \alpha[x]$$

The "source" term within the framework of the Kantarovitch' s method consists of *f1* -, *f2* - and *f3* - functions, which are integrated products of the *Q*-term and the variational derivatives of *A1*:

```
f1 = Integrate[Expand[Q * u1], {t, -∞, ∞}, Assumptions → t ∈ Reals && T[x] ∈ Reals &&
    T[x] > 0 && α[x] ∈ Reals && α[x] > 0 && ϕ[x] ∈ Reals && ξ > 0 && ζ ∈ Reals]
f2 = Integrate[Expand[Q * u2], {t, -∞, ∞}, Assumptions → t ∈ Reals && T[x] ∈ Reals &&
    T[x] > 0 && α[x] ∈ Reals && α[x] > 0 && ϕ[x] ∈ Reals && ξ > 0 && ζ ∈ Reals]
f3 = Integrate[Expand[Q * u3], {t, -∞, ∞}, Assumptions → t ∈ Reals && T[x] ∈ Reals &&
    T[x] > 0 && α[x] ∈ Reals && α[x] > 0 && ϕ[x] ∈ Reals && ξ > 0 && ζ ∈ Reals]
```

$$-\frac{2}{15}\, i\, \alpha[x] \left(15\,\Gamma\, T[x] - 10\,\kappa\, T[x]\, \alpha[x]^2 + 8\,\zeta\,\kappa\, T[x]\, \alpha[x]^4 + \frac{5\,\rho\left(\tau - 3\, T[x]^2\right)}{T[x]\left(1 + 2\,\sigma\, T[x]\, \alpha[x]^2\right)}\right)$$

$$-\frac{1}{45}\, i\, \alpha[x]^2 \left(45\,\Gamma - 15\,\kappa\,\alpha[x]^2 + 8\,\zeta\,\kappa\,\alpha[x]^4 - \frac{15\,\rho\left(\tau + 3\, T[x]^2\right)}{T[x]^2\left(1 + 2\,\sigma\, T[x]\, \alpha[x]^2\right)}\right)$$

$$\frac{2}{15}\,\alpha[x]^2 \left(-15\,\Gamma\, T[x] + 10\,\kappa\, T[x]\, \alpha[x]^2 - 8\,\zeta\,\kappa\, T[x]\, \alpha[x]^4 - \frac{5\,\rho\left(\tau - 3\, T[x]^2\right)}{T[x]\left(1 + 2\,\sigma\, T[x]\, \alpha[x]^2\right)}\right)$$

Thus, the "driven" Euler-Lagrange equations are (NB: the right-hand sides, or "source" terms, are **2 Re[*f*]**):

```
fun1 = eq1[[1]] == 0
fun2 = eq2[[1]] == 0
fun3 = eq3[[1]] == 2 f3
```

$$\frac{2\,\alpha[x]\left(\beta - 2\, T[x]^2\left(2\,\gamma\,\alpha[x]^2 + 3\,\phi'[x]\right)\right)}{3\, T[x]} == 0$$

$$-\frac{\alpha[x]^2\left(\beta + 2\, T[x]^2\left(\gamma\,\alpha[x]^2 + 3\,\phi'[x]\right)\right)}{3\, T[x]^2} == 0$$

$$2\,\alpha[x]\,(\alpha[x]\, T'[x] + 2\, T[x]\,\alpha'[x]) == \frac{4}{15}\,\alpha[x]^2\left(-15\,\Gamma\, T[x] + 10\,\kappa\, T[x]\,\alpha[x]^2 - 8\,\zeta\,\kappa\, T[x]\,\alpha[x]^4 - \frac{5\,\rho\left(\tau - 3\, T[x]^2\right)}{T[x]\left(1 + 2\,\sigma\, T[x]\,\alpha[x]^2\right)}\right)$$

Soliton (steady-state chirp-free pulse)

The soliton parameters are x-independent in this case (except the phase ϕ)

```
fun3 /. {α′[x] → 0, T′[x] → 0}
fun3b = (((%[[2]]) / 2 / α[x]) /. {α[x] → α, T[x] → T}) == 0
Solve[{(fun1 /. {α[x] → α, T[x] → T}), (fun2 /. {α[x] → α, T[x] → T})}, {ϕ′[x], T}]
```

$$0 == \frac{4}{15}\,\alpha[x]^2\left(-15\,\Gamma\,T[x] + 10\,\kappa\,T[x]\,\alpha[x]^2 - 8\,\zeta\,\kappa\,T[x]\,\alpha[x]^4 - \frac{5\,\rho\,\left(\tau - 3\,T[x]^2\right)}{T[x]\,\left(1 + 2\,\sigma\,T[x]\,\alpha[x]^2\right)}\right)$$

$$\frac{2}{15}\,\alpha\left(-15\,T\,\Gamma + 10\,T\,\alpha^2\,\kappa - 8\,T\,\alpha^4\,\zeta\,\kappa - \frac{5\,\rho\,\left(-3\,T^2 + \tau\right)}{T\,\left(1 + 2\,T\,\alpha^2\,\sigma\right)}\right) == 0$$

$$\left\{\left\{\phi'[x] \to -\frac{\alpha^2\,\gamma}{2},\ T \to -\frac{\sqrt{\beta}}{\alpha\,\sqrt{\gamma}}\right\},\ \left\{\phi'[x] \to -\frac{\alpha^2\,\gamma}{2},\ T \to \frac{\sqrt{\beta}}{\alpha\,\sqrt{\gamma}}\right\}\right\}$$

Thus, one has solutions for the phase slip $\partial_x \phi$ and the soliton width T. The equation for the soliton amplitude remains

```
fun3b
```

$$\frac{2}{15}\,\alpha\left(-15\,T\,\Gamma + 10\,T\,\alpha^2\,\kappa - 8\,T\,\alpha^4\,\zeta\,\kappa - \frac{5\,\rho\,\left(-3\,T^2 + \tau\right)}{T\,\left(1 + 2\,T\,\alpha^2\,\sigma\right)}\right) == 0$$

The crucial step is to renormalize the inverse squared gain-bandwidth to that for a saturated gain. That is $\tau_{\text{new}} = \frac{\rho\,\tau_{\text{old}}}{1 + 2\,\alpha^2\,T\,\sigma}$. Then, let us introduce the saturated net gain coefficient $\Sigma = \frac{\rho}{1+2\,\alpha^2\,T\,\sigma} - \Gamma < 0$. This term can be expanded in the vicinity of 0 (see previous section). Thus, the modified equation is

$$\text{fun3c} = \text{FullSimplify}\left[\left(\left(3\,T\,\Sigma - \frac{\tau}{T} + 2\,\kappa\,T\,\alpha^2\left(1 - \frac{4}{5}\,\zeta\,\alpha^2\right)\right)\ /.\ T \to \frac{\sqrt{\beta}}{\alpha\,\sqrt{\gamma}}\right)\right] == 0$$

$$\frac{-8\,\alpha^4\,\beta\,\zeta\,\kappa + 15\,\beta\,\Sigma + 5\,\alpha^2\,\left(2\,\beta\,\kappa - \gamma\,\tau\right)}{5\,\alpha\,\sqrt{\beta}\,\sqrt{\gamma}} == 0$$

- Let’s the power to be normalized to ζ and $c=\tau\gamma/\beta\kappa$ is some new control parameter. Ξ is the cw-energy normalized to $\sqrt{\kappa\,\zeta/\tau}$, and the pulse width is normalized to $\sqrt{\kappa/\zeta\tau}$. Then (P is the power)

```
fun3d = -8 P^2 β ζ κ + 15 β Σ + 5 P (2 β κ - γ τ) == 0
fun3e = -8 P^2 + 15 Σ ζ/κ + 5 P (2 - c) == 0
Solve[fun3e, P]
solP1 = 5/16 (2 - c - Sqrt[(c - 2)^2 + 96/5 ζ Σ/κ])
solP2 = 5/16 (2 - c + Sqrt[(c - 2)^2 + 96/5 ζ Σ/κ])
FullSimplify[(2 solP1/ζ (T /. T -> Sqrt[β]/(α Sqrt[γ]))) /. α -> Sqrt[solP1/ζ]]
```

$$\Xi 1 = \frac{\sqrt{5}\sqrt{2-c+\sqrt{(-2+c)^2+\frac{96\zeta\Sigma}{5\kappa}}}}{2\sqrt{c}}$$

$$\Xi 2 = \frac{\sqrt{5}\sqrt{2-c-\sqrt{(-2+c)^2+\frac{96\zeta\Sigma}{5\kappa}}}}{2\sqrt{c}}$$

$$-8P^2\beta\zeta\kappa+15\beta\Sigma+5P(2\beta\kappa-\gamma\tau)==0$$

$$5(2-c)P-8P^2+\frac{15\zeta\Sigma}{\kappa}==0$$

$$\left\{\left\{P\to\frac{10\kappa-5c\kappa-\sqrt{(-10\kappa+5c\kappa)^2+480\zeta\kappa\Sigma}}{16\kappa}\right\},\left\{P\to\frac{10\kappa-5c\kappa+\sqrt{(-10\kappa+5c\kappa)^2+480\zeta\kappa\Sigma}}{16\kappa}\right\}\right\}$$

$$\frac{5}{16}\left(2-c-\sqrt{(-2+c)^2+\frac{96\zeta\Sigma}{5\kappa}}\right)$$

$$\frac{5}{16}\left(2-c+\sqrt{(-2+c)^2+\frac{96\zeta\Sigma}{5\kappa}}\right)$$

$$\frac{\sqrt{\beta}\sqrt{-\frac{-10+5\,c+\sqrt{25\,(-2+c)^2+\frac{480\,\zeta\,\Sigma}{\kappa}}}{\zeta}}}{2\sqrt{\gamma}}$$

$$\frac{\sqrt{5}\sqrt{2-c+\sqrt{(-2+c)^2+\frac{96\,\zeta\,\Sigma}{5\,\kappa}}}}{2\sqrt{c}}$$

$$\frac{\sqrt{5}\sqrt{2-c-\sqrt{(-2+c)^2+\frac{96\,\zeta\,\Sigma}{5\,\kappa}}}}{2\sqrt{c}}$$

One can see that two solutions (two soliton branches) exist.
The solution for the saturated net-gain is

```
Solve[(-2 + c)^2 + (96 ζ Σ)/(5 κ) == 0, Σ][[1]][[1]]
```

$$\Sigma \to -\frac{5\,(-2+c)^2\,\kappa}{96\,\zeta}$$

that gives the normalized energy curve separated both branches:

```
FullSimplify[Ξ1 /. Σ → -(5 (-2 + c)^2 κ)/(96 ζ)]
```

$$\frac{\sqrt{5}\sqrt{2-c}}{2\sqrt{c}}$$

These solutions allow plotting the two-dimensional representation of the soliton parametric space (master diagram). Such a diagram is shown below in linear and logarithmic scales for Σ=0 and -0.005.

- The curve $\Xi=\sqrt{\frac{5}{c}}$ gives a good zero-gain asymptotic for $E\to\infty$.

$$\texttt{Plot}\Big[\Big\{\Xi1\ /.\ \Sigma\to 0,\ \frac{1}{2}\sqrt{\frac{5\ (2-c)}{c}}\ ,\ \Xi1\ /.\ \{\kappa\to 0.04,\ \xi\to 0.04/4/0.015,\ \Sigma\to -0.005\},$$

$$\Xi2\ /.\ \{\kappa\to 0.04,\ \xi\to 0.04/4/0.015,\ \Sigma\to -0.005\},\ \sqrt{\frac{5}{c}}\ \Big\},$$

$$\{c,\ 0,\ 2\},\ \texttt{AxesLabel}\to\{c,\ \texttt{Energy}\}\Big]$$

$$\texttt{Plot}\Big[\Big\{\texttt{Log10}[\Xi1]\ /.\ \{\Sigma\to 0,\ c\to 10^{x}\},\ \texttt{Log10}\Big[\frac{1}{2}\sqrt{\frac{5\ (2-c)}{c}}\ \Big]\ /.\ c\to 10^{x},$$

$$\texttt{Log10}[\Xi1]\ /.\ \{c\to 10^{x},\ \kappa\to 0.04,\ \xi\to 0.04/4/0.015,\ \Sigma\to -0.005\},$$
$$\texttt{Log10}[\Xi2]\ /.\ \{c\to 10^{x},\ \kappa\to 0.04,\ \xi\to 0.04/4/0.015,\ \Sigma\to -0.005\},$$

$$\texttt{Log10}\Big[\sqrt{\frac{5}{c}}\ \Big]\ /.\ c\to 10^{x}\Big\},\ \{x,\ -5,\ 1\},\ \texttt{AxesLabel}\to\{\texttt{log10}[c],\ \texttt{log10}[\texttt{Energy}]\}\Big]$$





It is interesting, that the master diagram for the chirp-free soliton resembles that for the chirped dissipative soliton developing in the normal dispersion regime [V.L.Kalashnikov, et al., Applied Physics B **83**, 503 (2006) but there exists no the dissipative soliton resonance [W.Chang, et al., Phys. Rev. A 78, 023830 (2008)] for a chirp-free soliton.

- Hence, the asymptotic scaling law is

$$E \approx \sqrt{\frac{20\,\mu\beta}{\kappa\gamma}} = \sqrt{\frac{5\,\beta}{\xi\gamma}} \tag{2}$$

- and the corresponding pulse amplitude and width are

```
Solve[{2 α^2 T == Sqrt[(20 μ β)/(κ γ)], T == Sqrt[β]/(α Sqrt[γ])}, {α, T}] // FullSimplify
```

$$\left\{\left\{\alpha \to \frac{\sqrt{5}\,\sqrt{\gamma}\,\sqrt{\frac{\beta\,\mu}{\gamma\,\kappa}}}{\sqrt{\beta}},\ T \to \frac{\kappa\,\sqrt{\frac{\beta\,\mu}{\gamma\,\kappa}}}{\sqrt{5}\,\mu}\right\}\right\}$$

- NB: the asymptotic soliton width is defined by both nonlinear parameters and dispersion (not gain bandwidth).

The dimensionless soliton width evolves with c as

```
Plot[(Log10[1/(α Sqrt[c])] /. α -> Sqrt[solP2]) /. {Σ -> 0, c -> 10^x}, {x, -5, 1},
 AxesLabel -> {log10[c], log10[T]}, AxesLabel -> {lg[c], lg[T]}]
```



As one can see from the previous picture, the energy scaling mechanisms in for the cubic-quintic nonlinear Ginzburg-Landau is similar for both anomalous and normal dispersion regimes: the scaling is provided by pulse stretching. The peak power is fixed by the quntic term provided the saturation of nonlinear gain:

```
Plot[solP2 /. {Σ → 0, c → 10^x}, {x, -5, 1}, AxesLabel → {log10[c], P}]
```



Rough estimation of the self-amplitude modulation parameters

The self-amplitude modulation parameters cannot be calculated “ab ovo” but some rough estimation can be obtained analytically [H.A. Haus, et al., IEEE J. Quantum Electr. **28**, 2086 (1992); K.-H. Lin, W.-F. Hsieh, J. Opt. Soc. Am. B **11**, 737 (1994); Sh. Yefet, A. Pe’er, Appl. Sci. **3**, 694 (2013)]. We will consider a simplest monolithic Kerr-lens mode locking system, which fits for distributed model under consideration. It can be reduced to a free-space propagation model for a Gaussian beam by rescaling of the imaginary part of q^{-1}-parameter:

$$F1 = \frac{1}{q_0} == -i \frac{\lambda}{\pi w_0^2} \sqrt{1-K}$$

$$\frac{1}{q_0} == -\frac{i \sqrt{1-K}\, \lambda}{\pi w_0^2}$$

where $K=\frac{P}{P_{cr}}$, P_{cr} is the critical power of self-focusing and w_0 is the waist size on input (plane wave).

Then, the quasi-free-space propagation on the distance z results in a new q-parameter, which equals to

$$F2 = (q_0 + z) \;/.\; q_0 \to -\frac{i \pi w_0^2}{\sqrt{1-K}\, \lambda}$$

$$z - \frac{i \pi w_0^2}{\sqrt{1-K}\, \lambda}$$

The imaginary part of q^{-1} has to be inversely rescaled:

```
Numerator[1 / F2] * (z + ⅈ π w0^2 / (Sqrt[1 - K] λ))
FullSimplify[Denominator[1 / F2] * (z + ⅈ π w0^2 / (Sqrt[1 - K] λ))]
F3 = Expand[%% / %]
```

$$z + \frac{i \pi w_0^2}{\sqrt{1 - K}\, \lambda}$$

$$z^2 - \frac{\pi^2 w_0^4}{(-1 + K)\, \lambda^2}$$

$$\frac{z}{z^2 - \frac{\pi^2 w_0^4}{(-1+K)\, \lambda^2}} + \frac{i \pi w_0^2}{\sqrt{1 - K}\, \lambda \left(z^2 - \frac{\pi^2 w_0^4}{(-1+K)\, \lambda^2}\right)}$$

```
F4 = π w0^2 / (Sqrt[1 - K] λ (z^2 - π^2 w0^4 / ((-1 + K) λ^2))) / Sqrt[1 - K]
```

$$\frac{\pi w_0^2}{(1 - K)\, \lambda \left(z^2 - \frac{\pi^2 w_0^4}{(-1+K)\, \lambda^2}\right)}$$

Then, the new imaginary part of q-parameter is:

```
F5 = π w^2 / λ == FullSimplify[1 / F4]
```

$$\frac{\pi w^2}{\lambda} == -\frac{(-1 + K)\, z^2 \lambda}{\pi w_0^2} + \frac{\pi w_0^2}{\lambda}$$

Hence, a new squared beam size is

```
FullSimplify[Solve[F5 /. w^2 → x, x]][[1]][[1]] /. x → w^2
```

$$w^2 \rightarrow -\frac{(-1 + K)\, z^2 \lambda^2}{\pi^2 w_0^2} + w_0^2$$

The loss on aperture can be estimated as $L = e^{-D^2/w^2}$, where D is the aperture size and

```
F6 = D^2/w^2 == D^2/((1-K) z^2 λ^2/(π^2 w0^2) + w0^2)
```

$$\frac{D^2}{w^2} == \frac{D^2}{\frac{(1-K)\, z^2 \lambda^2}{\pi^2 w_0^2} + w_0^2}$$

After some manipulations, one has

```
Collect[Expand[Denominator[F6[[2]]]], K]
Coefficient[%, K] (*power-dependent part*)
F7 = Coefficient[%%, K, 0](*power-independent part*)
F8 = 1 + %% * K / F7
```

$$\frac{z^2 \lambda^2}{\pi^2 w_0^2} - \frac{K z^2 \lambda^2}{\pi^2 w_0^2} + w_0^2$$

$$-\frac{z^2 \lambda^2}{\pi^2 w_0^2}$$

$$\frac{z^2 \lambda^2}{\pi^2 w_0^2} + w_0^2$$

$$1 - \frac{K z^2 \lambda^2}{\pi^2 w_0^2 \left(\frac{z^2 \lambda^2}{\pi^2 w_0^2} + w_0^2\right)}$$

$F9 = F7 / D^2$ can be considered as the saturable loss coefficient (modulation depth) μ

```
F9 = w0^2 * Expand[F7 / w0^2] / D^2 (*μ-coefficient*)
```

$$\frac{\left(1 + \frac{z^2 \lambda^2}{\pi^2 w_0^4}\right) w_0^2}{D^2}$$

This coefficient corresponds to a linear beam propagation and, since z-distance from a nonlinear medium to an aperture is quite large (i.e. the second term in denominator >>1) one may estimate the modulation depth as

```
w0^2 Numerator[F9][[1]][[2]] / D^2 (*modulation depth μ*)
```

$$\frac{z^2 \lambda^2}{D^2 \pi^2 w_0^2}$$

- which is inversely proportional to w_0^{-2}. That is a thin waist causes a larger beam diffraction.

The nonlinear contribution to a loss coefficient can be estimated as (a nonlinear medium is thin and one may

neglect the diffraction inside it):

```
-w0^2/D^2 * F8[[2]]
Expand[Denominator[%]] /. z → 0
Numerator[%%] / %
```

$$\frac{K z^2 \lambda^2}{D^2 \pi^2 \left(\frac{z^2 \lambda^2}{\pi^2 w_0^2} + w_0^2\right)}$$

$$D^2 \pi^2 w_0^2$$

$$\frac{K z^2 \lambda^2}{D^2 \pi^2 w_0^2}$$

- Hence, a self-amplitude modulation contribute as $\propto w_0^{-2}$

Conclusion

The scaling laws for two types of self-amplitude modulation in a model of Kerr-lens mode locked laser operating in anomalous dispersion regime have been obtained:

$E \approx \frac{3.5 \sqrt{\mu}\, \beta}{\sqrt{\tau}\, \gamma}$, $T \approx \frac{0.57 \sqrt{\tau}}{\sqrt{\mu}}$, $P \approx \frac{3 \beta \mu}{\gamma \tau}$ **for a perfectly saturable absorber**

and

$E \approx \sqrt{\frac{20\, \mu \beta}{\kappa \gamma}} = \sqrt{\frac{5 \beta}{\zeta \gamma}}$, $T \approx \sqrt{\frac{\beta \kappa}{5 \gamma \mu}}$, $P \approx \frac{5 \mu}{\kappa}$ **for cubic-quintic model of a saturable absorber**

For both types of models, both energy and pulse width scale with a waist size in a nonlinear medium as $\propto \boldsymbol{w_0}$ (one has note that γ scales as w_0^2). An asymptotical power is not beam size dependent for both models.
Main differences can be classified in the following ways:
1) energy scales with dispersion more rapidly for a perfectly saturable absorber; spectral dissipation is irrelevant for the energy scaling in a cubic-quintic model
2) pulse width scaling is defined by bandwidth/dispersion for perfectly saturable/cubic-quintic models, respectively
3) asymptotic power for a cubic-quintic model is defined by nonlinear gain saturation. However, it is defined by the combined action of dispersion, self-phase modulation and spectral dissipation for a perfectly saturable absorber model.